Part III:

Programs

Introduction

This part contains hard copy of various programs for graphing calculators. The following calculators are considered:

- Texas Instruments TI-82
- Texas Instruments TI-83
- Texas Instruments TI-85
- Texas Instruments TI-86
- Texas Instruments TI-89
- Casio CFX-9800G
- Casio CFX-9850G
- Hewlett-Packard HP-38G

Information on how to enter these programs manually and how to run them is provided for each of these calculators at the beginning of each section. However, these short instructions do not claim to tell the whole story on how to operate the calculator; the reader is directed to the Owner's Manuals for the various calculators if more detailed information is desired.

Since the TI-82 and TI-83 are very similar machines, the same programs will work on either calculator; these calculators are discussed in a single section. Likewise the similarity of the TI-85 and TI-86 allows the same programs to work on either of them, and these calculators are covered in one section also. The programs for the Texas Instruments calculators are also available via the World Wide Web; see the Texas Instruments sections for more details.

The author of this part wishes to acknowledge the large amount of help he received from two people in the writing of these programs. Thomas Hungerford wrote many of the programs for the Texas Instruments calculators, while Alan Ziv provided valuable expertise and wrote several of the programs for the Casio calculators.

Programs for the TI-82 and TI-83

Introduction

The following section contains programs for the TI-82 and TI-83 calculators. These programs allow the calculator to do various financial calculations, linear programming, numerical integration, Euler's method for the numerical solution of differential equations, and to find probabilities associated with the normal distribution.

Entering the Programs Manually: The Program Editor is used to enter a program manually. To access the Program Editor press [PGRM]. To enter a new program, select NEW and press [ENTER]; to edit an existing program, select EDIT, then move down the list of programs to the program you wish to edit. If you are creating a new program, you will first be prompted for a program name. After entering the name, you will see a colon with a cursor to its left. You are now to enter the first line of the program. Each line of a program begins with a colon; these are supplied by the Editor.

When entering a program from the Program Editor, simple commands may be entered as they appear on the keyboard. Alphabet keys are also easily accessible from the keyboard. Only upper-case letters are available. To enter a letter, press [ALPHA] then the appropriate key. Pressing [2nd] [ALPHA] locks the calculator in alphabet mode. You can easily tell when the calculator is in alphabet mode by the cursor, which flashes an A when in alphabet mode. The $\rightarrow$ symbol which you will need for the programs is the [STO▶] key. In addition to keyboard symbols, you will need to find the commands such as ClrHome, Disp, and round which appear in the programs. These commands are located in various menus; for example ClrHome and Disp are in the I/O submenu of the [PRGM] menu in the Program Editor. So to enter ClrHome from the Program Editor, you would press [PGRM], then select the I/O menu, then move down that menu until you reach 8:ClrHome. Pressing [ENTER] now will place ClrHome at the point where you left the Program Editor. Some commands are somewhat difficult to find. On the TI-82, you will need to use the Table of Functions and Instructions in Appendix A of the Guidebook to help you find the proper menus. On the TI-83, there is another way to enter commands into the program. Pressing [2nd] [0] accesses the CATALOG menu. Pressing a letter key now will send you to the commands beginning with that letter. Pressing [ENTER] will place the selected command into your program. So to enter ClrHome by this method, you would enter the catalog, press [PRGM] to access the letter C (the calculator is already in ALPHA mode), then move down the list until the cursor points at ClrHome, then press [ENTER].

When you are finished entering the program, press [2nd] [MODE] (QUIT) to exit the Program Editor. The program is now stored under the name you have given it.

Entering the Programs from another TI-82 or TI-83: Programs and other data may be transferred from a TI-82 to a TI-82 or from a TI-83 to a TI-83. It is also possible to transfer files from a TI-82 to a TI-83, although there are differences between the calculators which could lead to errors in programs transferred in this manner. To perform a transfer you need the communcation link cable which was enclosed with your calculator. Directions on how to perform this process are given in Chapter 16 of the TI-82 Guidebook and in Chapter 19 of the TI-83 Guidebook.

Entering the Programs from a Computer: The programs in this section are available on the World Wide Web at www.aw.com/LGR. The names of these programs are given at the beginning of each program listing. The programs may be downloaded to a PC then transferred to your calculator. You will need the TI-GRAPH LINK software and cable to perform this transfer. Directions on transferring programs from a PC to the calculator are included with the TI-GRAPH LINK.

Running the Programs: To run a program, press [PGRM], make sure that EXEC is selected at the top of the screen, then move down the list of programs to the one you wish to run. Pressing [ENTER] will clear the screen and the name of the program will appear. Pressing [ENTER] again begins execution of the program.

Trapezoidal Rule – TZOID.82P

The following program uses the Trapezoidal Rule to approximate the value of a definite integral. The function you wish to integrate must be stored in the variable Y_1 before you execute this program. You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N).

```
:ClrHome
:Prompt A
:Prompt B
:Prompt N
:(B-A)/N → D
:0 → S
:Y1(A) → S
:Y1(B)+S → S
:For(K,1,N-1,1)
:2Y1(A+K*D)+S → S
:End
:Disp S*D/2
```

Simpson's Rule – SIMPSON.82P

The following program uses Simpson's Rule to approximate the value of a definite integral. The function you wish to integrate must be stored in the variable Y_1 before you execute this program. You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N). You must choose an even number for N.

```
:ClrHome
:Prompt A
:Prompt B
:Prompt N
:(B-A)/N → D
:0 → S
:Y1(A) → S
:Y1(B)+S → S
:For(K,1,N/2,1)
:4Y1(A+(2K-1)*D)+S → S
:End
:For(K,1,N/2-1,1)
:2Y1(A+2K*D)+S → S
:End
:Disp S*D/3
```

Integration by Endpoints or Midpoint – LSUM.82P, RSUM.82P, and MSUM.82P

The following programs approximate the value of a definite integral. The function you wish to integrate must be stored in the variable Y_1 before you execute this program. You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N).

Left Endpoints – LSUM.82P:

```
:ClrHome
:Prompt A
:Prompt B
:Prompt N
:(B-A)/N → D
```

```
:0 → S
:For(K,0,N-1,1)
:Y1(A+K*D)+S → S
:End
:Disp S*D
```

Right Endpoints – RSUM.82P:

```
:ClrHome
:Prompt A
:Prompt B
:Prompt N
:(B-A)/N → D
:0 → S
:For(K,1,N,1)
:Y1(A+K*D)+S → S
:End
:Disp S*D
```

Midpoints – MSUM.82P:

```
:ClrHome
:Prompt A
:Prompt B
:Prompt N
:(B-A)/N → D
:0 → S
:For(K,1,N,1)
:Y1(A+(2K-1)*D/2)+S → S
:End
:Disp S*D
```

Euler's Method – EULER.82P

The following program performs Euler's method to approximate the solution to the differential equation

$$\frac{dy}{dx} = f(x, y).$$

The function $f(x, y)$ must be stored as Y_1 before you execute this program, and each occurence of y in $f(x, y)$ should be entered as Y. You also must enter an initial point (both x and y coordinates), the increment between successive x values, and the value of x at which you wish to find an estimate for y.

```
:ClrHome
:Input "INITIAL X VALUE?",X
:Input "INITIAL Y VALUE?",Y
:Input "INCREMENT?",H
:Input "FINAL X VALUE?",Z
:While X<Z
:Y+H*Y1 → Y
:X+H → X
:End
:Disp "FINAL (X,Y)"
:Disp X,Y
```

Normal Probability – NRML.82P

The following program computes the probability that a normal random variable with given mean and standard deviation lies between two given values. The probability that the random variable lies below a particular value may be computed by inputting -1E99 for the lower limit; the probability that the random variable lies above a particular value may be computed by inputting 1E99 for the upper limit. The TI-83 has a built-in program called `normalcdf` which performs the work of this program. The program `normalcdf` is located in the `DISTR` menu; see the TI-83 Guidebook for more details.

```
:ClrHome
:Input "MEAN?",M
:Input "STD DEV?",S
:Input "LOWER LIMIT?",A
:(A-M)/S→A
:If A<-10
:-10 → A
:If A>10
:10 → A
:Input "UPPER LIMIT?",B
:(B-M)/S→B
:If B<-10
:-10 → B
:If B>10
:10 → B
:Disp "PR(A<N(M,S)<B)="
:Disp round(fnInt(e^(X^2/-2),X,A,B)/√(2*π),4)
```

Programs for the TI-85 and TI-86

Introduction

The following section contains programs for the TI-85 and TI-86 calculators. These programs allow the calculator to do various financial calculations, linear programming, numerical integration, Euler's method for the numerical solution of differential equations, and to find probabilities associated with the normal distribution.

Entering the Programs Manually: The Program Editor is used to enter a program manually. To access the Program Editor press PGRM. To enter a new program, select EDIT; to edit an existing program, select EDIT, press the menu key under the name of the program you wish to edit, then press ENTER. The programs are listed in alphabetical order; if you don't see the program you want, try pressing MORE to expose the next portion of the list. If you are creating a new program, you will first be prompted for a program name. After entering the name, you will see a colon with a cursor to its left. You are now to enter the first line of the program. Each line of a program begins with a colon; these are supplied by the Editor.

When entering a program, simple commands may be entered as they appear on the keyboard. Alphabet keys are also easily accessible from the keyboard. To enter an upper case letter, press ALPHA then the appropriate key. To enter a lower case letter, press 2nd ALPHA then the appropriate key. Pressing ALPHA ALPHA locks the calculator in upper-case alphabet mode; pressing 2nd ALPHA ALPHA locks the calculator in lower-case alphabet mode. You can easily tell when the calculator is in alphabet mode by the cursor, which flashes an A when in upper-case alphabet mode and an a when in lower-case alphabet mode. The $\rightarrow$ symbol which you will need for these programs is the STO▶ key. In addition to keyboard symbols, you will need to find the commands such as ClLCD, Disp, and round which appear in the programs. These commands are located in various menus. For example, ClLCD and Disp are in the I/O menu, which is accessed by pressing F3 when entering a program. So to enter ClLCD, you would select the I/O menu by pressing F3, then move across that menu (pressing MORE to get more of the menu) until you reach ClLCD. Pressing the menu key beneath ClLCD now will place ClLCD at the point where you left the editor. Some commands are somewhat difficult to find. There is another way to enter commands into the program. Pressing 2nd CUSTOM will access the CATALOG menu. On the TI-85 the catalog will appear automatically; on the TI-86 you must additionally press F1. Pressing a letter key now will send you to the commands beginning

with that letter. Pressing [ENTER] will place the selected command into your program. So to enter ClLCD by this method, you would enter the catalog, press [COS] to enter a C (the calculator is already in ALPHA mode), move down the list until the cursor points at ClLCD, then press [ENTER].

When you are finished entering the program, press [2nd] [EXIT] (QUIT) to exit the editor. The program is now stored under the name you have given it.

Entering the Programs from another TI-85 or TI-86: Programs and other data may be transferred from a TI-85 to a TI-85 or from a TI-86 to a TI-86. It is also possible to transfer programs from a TI-85 to a TI-86, although there are differences between the calculators which could lead to errors in programs transferred in this manner. To perform a transfer you need the communcation link cable which was enclosed with your calculator. Directions on how to perform this process are given in Chapter 19 of the TI-85 Guidebook and in Chapter 18 of the TI-86 Guidebook.

Entering the Programs from a Computer: The programs in this section are available on the World Wide Web at www.aw.com/LGR. The names of these programs are given at the beginning of each program listing. The programs may be downloaded to a PC then transferred to your calculator. You will need the TI-GRAPH LINK software and cable to perform this transfer. Directions on transferring programs from a PC to the calculator are included with the TI-GRAPH LINK.

Running the Programs: To run a program, press [PGRM] then [F1] to access the NAMES submenu. Select the name of the program as you would for editing. Pressing [ENTER] will cause the name of the program to appear; pressing [ENTER] again begins execution of the program.

Trapezoidal Rule – TZOID.85P

The following program uses the Trapezoidal Rule to approximate the value of a definite integral. The function you wish to integrate must be stored in the variable Y_1 before you execute this program. You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N).

```
:ClLCD
:Prompt A
:Prompt B
:Prompt N
:(B-A)/N → D
:0 → S
:A → x
:y1 → S
:B → x
:y1+S → S
:For(K,1,N-1,1)
```

```
:A+K*D → x
:2y1+S → S
:End
:Disp S*D/2
```

Simpson's Rule – SIMPSON.85P

The following program uses Simpson's Rule to approximate the value of a definite integral. The function you wish to integrate must be stored in the variable y_1 before you execute this program. You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N). You must choose an even number for N.

```
:ClLCD
:Prompt A
:Prompt B
:Prompt N
:(B-A)/N → D
:0 → S
:A → x
:y1 → S
:B → x
:y1+S → S
:For(K,1,N/2,1)
:A+(2K-1)*D → x
:4y1+S → S
:End
:For(K,1,N/2-1,1)
:A+2K*D → x
:2y1+S → S
:End
:Disp S*D/3
```

Integration by Endpoints or Midpoint – LSUM.85P, RSUM.85P, and MSUM.85P

The following programs approximate the value of a definite integral. The function you wish to integrate must be stored in the variable y_1 before you execute this program. You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N).

Left Endpoints – LSUM.85P:

```
:ClLCD
:Prompt A
:Prompt B
:Prompt N
:(B-A)/N → D
:0 → S
:For(K,0,N-1,1)
:A+K*D → x
:y1+S → S
:End
:Disp S*D
```

Right Endpoints – RSUM.85P:

```
:ClLCD
:Prompt A
:Prompt B
:Prompt N
:(B-A)/N → D
:0 → S
:For(K,1,N,1)
:A+K*D → x
:y1+S → S
:End
:Disp S*D
```

Midpoints – MSUM.85P:

```
:ClLCD
:Prompt A
:Prompt B
:Prompt N
:(B-A)/N → D
:0 → S
:For(K,1,N,1)
:A+(2K-1)*D/2 → x
:y1+S → S
:End
:Disp S*D
```

Euler's Method – EULER1.85P

The following program performs Euler's method to approximate the solution to the differential equation

$$\frac{dy}{dx} = f(x, y).$$

The function $f(x)$ must be stored as $\mathtt{y}_1$ before you execute this program, and each occurence of y in $f(x, y)$ should be entered as Y. You also must enter an initial point (both x and y coordinates), the increment between successive x values, and the value of x at which you wish to find an estimate for y. The TI-86 has a program which graphs an approximate solution for a differential equation using Euler's method; see Chapter 10 of the TI-86 Guidebook for more details.

```
:ClLCD
:Input "INITIAL X VALUE?",x
:Input "INITIAL Y VALUE?",Y
:Input "INCREMENT?",H
:Input "FINAL X VALUE?",Z
:While x<Z
:Y+H*y1 → Y
:x+H → x
:End
:Disp "FINAL (X,Y)"
:Disp x,Y
```

Normal Probability – NRML.85P

The following program computes the probability that a normal random variable with given mean and standard deviation lies between two given values. The probability that the random variable lies below a particular value may be computed by inputting -1E99 for the lower limit; the probability that the random variable lies above a particular value may be computed by inputting 1E99 for the upper limit.

```
:ClLCD
:Input "MEAN?",M
:Input "STD DEV?",S
:Input "LOWER LIMIT?",A
:(A-M)/S→A
:If A<-10
:-10 → A
:If A>10
:10 → A
```

```
:Input "UPPER LIMIT?",B
:(B-M)/S→B
:If B<⁻10
:⁻10 → B
:If B>10
:10 → B
:Disp "PR(A<N(M,S)<B)="
:Disp round(fnInt(eˆ(Xˆ2/⁻2),X,A,B)/√(2*π),4)
```

Programs for the TI-89

Introduction

The following section contains programs for the TI-89 calculator. These programs allow the calculator to do various financial calculations, linear programming, numerical integration, Euler's method for the numerical solution of differential equations, and to find probabilities associated with the normal distribution.

Entering the Programs Manually: The Program Editor is used to enter a program manually. To access the Program Editor press [APPS] 6. To enter a new program, select 3:New. You will first be prompted for a program name and for the name of the folder in which you wish to store the program. After entering the name, you will see the name of the program, the commands Pgrm and EndPgrm, and a colon with a blank line following it. You are to enter the first line of the program on that blank line. Each line of a program begins with a colon; these are supplied by the Editor. When you are done, pressing [2nd] [ESC] (QUIT) or [HOME] returns you to the home screen.The program is now stored under the name you have given it.

When entering a program, simple commands may be entered as they appear on the keyboard. Alphabet keys are also easily accessible from the keyboard. To enter an lower case letter, press [alpha] then the appropriate key. To enter a upper case letter, press [⇑] [ALPHA] then the appropriate key. Pressing [alpha] [alpha] or [a-lock] ([2nd] [alpha]) locks the calculator in alphabet mode. You can easily tell when the calculator is in alphabet mode by the icon a which appears in the status line at the bottom of the calculator screen. The → symbol which you will need for these programs is the [STO▶] key. In addition to keyboard symbols, you will need to find the commands such as ClrIO, Disp, and round which appear in the programs. These commands are located in various menus. For example, round is in the MATH Number menu, which is accessed by by pressing [2nd] [5], then selecting 1:Number. Pressing the right arrow key exposes a long list of commands. Moving down the list to 3:round(and pressing [ENTER] will place round(at the point where you left the Program Editor. The menus for some commands are somewhat difficult to find; all commands are listed in the catalog, which is accessed by pressing [CATALOG]. Pressing a letter key now will send you to the commands beginning with that letter, and pressing [ENTER] will place the selected command into your program. So to enter ClrIO by this method, you would enter the catalog, press [)] to enter a c (the calculator is

already in alphabet mode), move down the list until the cursor points at `ClrIO`, then press ENTER.

To edit an existing program, access the Program Editor and select `2:Open`. With the Variable box selected, press the right arrow key to see a list of programs in the current folder. Move the cursor down to the name of the program you wish to edit, then press ENTER twice. The programs are listed in alphabetical order; if you don't see the program you want, continue using the down arrow to expose more of the list.

Entering the Programs from another TI-89: Programs and other data may be transferred from a TI-89 to another TI-89. To perform a transfer you need the communication link cable which was enclosed with your calculator. Directions on how to perform this process are given in Chapter 22 of the TI-89 Guidebook.

Entering the Programs from a Computer: The programs in this section are available on the World Wide Web at `www.aw.com/LGR`. The names of these programs are given at the beginning of each program listing. The programs may be downloaded to your computer and then transferred to your calculator. You will need the TI-GRAPH LINK software and cable to perform this transfer. Directions on transferring programs from a PC to the calculator are included with TI-GRAPH LINK.

Running the Programs: To run a program, enter the name of the program on the entry line of the Home screen. Pressing ENTER begins execution of the program. For example, entering `fvann()` on the entry line, then pressing ENTER would start the first program listed below. You can also locate the program using the VAR-LINK feature on the calculator (2nd -) – see Chapter 21 of the TI-89 Guidebook for details.

Trapezoidal Rule – TZOID.89P

The following program uses the Trapezoidal Rule to approximate the value of a definite integral. The function you wish to integrate must be stored in the variable y_1 before you execute this program. You are also required to input the lower limit of integration (a), the upper limit of integration (b), and the number of subintervals (n).

```
:tzoid()
:Prgm
:ClrIO
:Prompt a
:Prompt b
```

```
:Prompt n
:(b-a)/n → d
:0 → s
:a → x
:y1(x) → s
:b → x
:y1(x)+s → s
:For k,1,n-1,1
:a+k*d → x
:2*y1(x)+s → s
:EndFor
:Disp s*d/(2.)
:EndPrgm
```

Simpson's Rule – SIMPSON.89P

The following program uses Simpson's Rule to approximate the value of a definite integral. The function you wish to integrate must be stored in the variable y_1 before you execute this program. You are also required to input the lower limit of integration (a), the upper limit of integration (b), and the number of subintervals (n). You must choose an even number for n.

```
:simpson()
:Prgm
:ClrIO
:Prompt a
:Prompt b
:Prompt n
:(b-a)/n → d
:0 → s
:a → x
:y1(x) → s
:b → x
:y1(x)+s → s
:For k,1,n/2,1
:a+(2*k-1)*d → x
:4*y1(x)+s → s
:EndFor
:For k,1,n/2-1,1
:a+2*k*d → x
:2*y1(x)+s → s
```

```
:EndFor
:Disp s*d/(3.)
:EndPrgm
```

Integration by Endpoints or Midpoint – LSUM.89P, RSUM.89P, MSUM.89P

The following programs approximate the value of a definite integral. The function you wish to integrate must be stored in the variable y_1 before you execute this program. You are also required to input the lower limit of integration (a), the upper limit of integration (b), and the number of subintervals (n).

Left Endpoints – LSUM.89P:

```
:lsum()
:Prgm
:ClrIO
:Prompt a
:Prompt b
:Prompt n
:(b-a)/n → d
:0 → s
:For k,0,n-1,1
:a+k*d → x
:y1(x)+s → s
:EndFor
:Disp s*d
:EndPrgm
```

Right Endpoints – RSUM.89P:

```
:rsum()
:Prgm
:ClrIO
:Prompt a
:Prompt b
:Prompt n
:(b-a)/n → d
:0 → s
```

```
:For k,1,n,1
:a+k*d → x
:y1(x)+s → s
:EndFor
:Disp s*d
:EndPrgm
```

Midpoints – MSUM.89P:

```
:msum()
:Prgm
:ClrIO
:Prompt a
:Prompt b
:Prompt n
:(b-a)/n → d
:0 → s
:For k,1,n,1
:a+(2*k-1)*d/2 → x
:y1(x)+s → s
:EndFor
:Disp s*d
:EndPrgm
```

Euler's Method – EULER.89P

The following program performs Euler's method to approximate the solution to the differential equation

$$\frac{dy}{dx} = f(x, y).$$

The function $f(x, y)$ must be stored as $\mathtt{y}_1$ before you execute this program, and each occurence of y in $f(x, y)$ should be entered as `y`. You also must enter an initial point (both x and y coordinates), the increment between successive x values, and the value of x at which you wish to find an estimate for y.

```
:euler()
:Prgm
:ClrIO
:Input "Initial x value?",x
:Input "Initial y value?",y
```

```
:Input "Increment?",h
:Input "Final x value?",z
:While x<z
:y+h*y1(x) → y
:x+h → x
:EndWhile
:Disp "Final (x,y)"
:Disp x,y
:EndPrgm
```

Normal Probability – NRML.89P

The following program computes the probability that a normal random variable with given mean and standard deviation lies between two given values. The probability that the random variable lies below a particular value may be computed by inputting $-\infty$ for the lower limit; the probability that the random variable lies above a particular value may be computed by inputting ∞ for the upper limit. The infinity sign can be found on the keyboard as ♦ CATALOG.

```
:nrml()
:Prgm
:ClrIO
:Input "Mean?",m
:Input "Standard deviation?",s
:Input "Lower limit?",a
:(a-m)/s→a
:Input "Upper limit?",b
:(b-m)/s→b
:Disp "Pr(a<N(m,s)<b)="
:Disp nInt(e^(⁻x^2/2),x,a,b)/√(2*π))
:EndPrgm
```

Programs for the Casio CFX-9800G

Introduction

The following section contains programs for the Casio CFX-9800G calculator. These programs allow the calculator to do various financial calculations, numerical intergration, Euler's method for the numerical solution of differential equations, and to find probabilities associated with the normal distribution.

Entering the Programs Manually: You should enter these programs using the File Editor Mode, which allows you to choose a name for each program. To enter File Editor Mode, select the PGRM icon from the main menu, then press F2, which is labelled EDT. Selecting F1 (labelled NEW) at this point will allow you to begin entering a program. You are first prompted for a filename; after inputting the filename press ENTER; you may then begin entering the program itself.

The program is entered by typing each line onto the calculator screen; to enter variable names and other data, a set of alphabet keys is provided. The ALPHA key activates the red versions of the keys, which are the letters of the alphabet and other symbols; SHIFT ALPHA locks the calculator in alphabet entry mode. Special symbols and programming commands are also needed. These may be placed into your program in three ways. First, the keyboard may be used to input the symbols marked on the keys. Second, pressing F6 (labelled SYM) in the initial editor gives access to other symbols. Third, programming symbols such as ? or $\Rightarrow$ may be entered by pressing SHIFT PRGM, then accessing the menus contained therein. Each line of the program is ended with the EXE key; the Editor will show a carriage return symbol at this point. These carriage returns are not shown in the program listings that follow.

Entering the Programs from another CFX-9800G: It is possible to download programs and other data from one CFX-9800G to another. The Casio SB-62 cable is used for these transfers. For details on how to perform such transfers, consult Chapter 13 in the Owner's Manual.

Running the Programs: To run the program while in File Editor Mode, simply select the name of the program from the list which appears when you enter the Mode. Pressing F4 (or RUN) will start the program. If you are not in File Editor Mode, the program may be run by pressing SHIFT PGRM F3 (labelled Prg). The calculator will ask for the program's name. The name must be entered

enclosed by double quotes; these are accessed by pressing SHIFT ALPHA F2. After entering the name, pressing EXE runs the program.

Trapezoidal Rule

The following program uses the Trapezoidal Rule to approximate the value of a definite integral. The function you wish to integrate must be stored in the Function Memory as f_6 before you execute this program. You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N).

```
"A="?→A
"B="?→B
"N="?→N
(B-A)÷N→D
N→K
0→S
A→X
Lbl 0
S+f6D→S
X+D→X
Dsz K
Goto 0
A→X
f6 →E
B→X
f6 →F
S+(F-E)D÷2→S
S
```

Simpson's Rule

The following program uses Simpson's Rule to approximate the value of a definite integral. The function you wish to integrate must be stored in the variable f_6 before you execute this program. You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N). You must choose an even number for N. You can also use the calculator's built-in numerical integration program, which uses Simpson's Rule as its algorithm. See the Owner's Manual for more information on the built-in function.

```
"A="?→A
"B="?→B
"N="?→N
(B-A)÷N → D
0 → S
A → X
f6 → S
B → X
f6+S → S
N÷2 → K
Lbl 0
A+(2K-1)D → X
S+4f6 → S
Dsz K
Goto 0
(N÷2)-1 → K
Lbl 1
A+2KD → X
S+2f6 → S
Dsz K
Goto 1
SD÷3
```

Integration by Endpoints or Midpoint

The following programs approximate the value of a definite integral. The function you wish to integrate must be stored in the Function Memory as f_6 before you execute this program. You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N).

Left Endpoints:

```
"A="?→A
"B="?→B
"N="?→N
(B-A)÷N→D
N→K
0→S
A→X
Lbl 0
```

```
S+f6D→S
X+D→X
Dsz K
Goto 0
S
```

Right Endpoints:

```
"A="?→A
"B="?→B
"N="?→N
(B-A)÷N→D
N→K
0→S
A→X
X+D→X
Lbl 0
S+f6D→S
X+D→X
Dsz K
Goto 0
S
```

Midpoints:

```
"A="?→A
"B="?→B
"N="?→N
(B-A)÷N→D
N→K
0→S
A→X
X+.5D→X
Lbl 0
S+f6D→S
X+D→X
Dsz K
Goto 0
S
```

Euler's Method

The following program performs Euler's method to approximate the solution to the differential equation

$$\frac{dy}{dx} = f(x).$$

The function $f(x)$ must be stored as $\mathtt{f}_6$ before you execute this program. You also must enter an initial point (both x and y coordinates), the increment between successive x values, and the value of x at which you wish to find an estimate for y.

```
"INITIAL X VALUE"?→X
"INITIAL Y VALUE"?→Y
"INCREMENT"?→H
"FINAL X VALUE"?→Z
Lbl 1
Y+H×f6 → Y
X+H → X
X<Z ⇒ Goto 1
"FINAL X VALUE"
X ▲
"FINAL Y VALUE"
Y
```

Normal Probability

The following program computes the probability that a normal random variable with given mean and standard deviation lies between two given values. The probability that the random variable lies below a particular value may be computed by inputting -1E99 for the lower limit; the probability that the random variable lies above a particular value may be computed by inputting 1E99 for the upper limit.

```
"MEAN"?→M
"STD DEV"?→S
"LOWER LIMIT"?→A
"UPPER LIMIT"?→B
(A-M)÷S→A
(B-M)÷S→B
Lbl 1
A<-10 ⇒ Goto 2
A>10 ⇒ Goto 3
Lbl 4
B<-10 ⇒ Goto 5
```

```
B>10 ⇒ Goto 6
"PR(A<N(M,S)<B)="
Fix 4
∫(e(Xˆ2÷-2),A,B)÷√(2×π) ▲
Lbl 2
-10 → A
Goto 1
Lbl 3
10 → A
Goto 1
Lbl 5
-10 → B
Goto 4
Lbl 6
10 → B
Goto 4
```

Programs for the Casio CFX-9850G

Introduction

The following section contains programs for the Casio CFX-9850G calculator. These programs allow the calculator to do various financial calculations, linear programming, numerical intergration, Euler's method for the numerical solution of differential equations, and to find probabilities associated with the normal distribution.

Entering the Programs Manually: To enter a program manually, select the PGRM icon from the main menu, then press F3, which is labelled NEW. You are first prompted for a program name; after inputting this name and pressing ENTER, you may begin entering the program itself.

The ALPHA key activates the red versions of the keys, which are the letters of the alphabet and other symbols; SHIFT ALPHA locks the calculator in alphabet entry mode. Special symbols and programming commands are also needed. These may be placed into your program in three ways. First, the keyboard may be used to input the symbols marked on the keys. Second, pressing F6 (SYBL) in the initial editor gives access to other symbols. Third, programming symbols such as ? or $\Rightarrow$ may be entered by pressing SHIFT PRGM, then accessing the menus contained therein. Each line of the program is ended with the EXE key; the Editor will show a carriage return symbol at this point. These carriage returns are not shown in the program listings that follow.

Entering the Programs from another CFX-9850G: It is possible to download programs and other data from one CFX-9850G to another. The Casio SB-62 cable is used for these transfers. For details on how to perform such transfers, consult Chapter 20 in the Owner's Manual.

Running the Programs: To run the program, first select the PGRM icon from the main menu, then select the name of the program from the list which appears. Pressing F1 (labelled EXE on the calculator screen) will start the program.

Trapezoidal Rule

The following program uses the Trapezoidal Rule to approximate the value of a definite integral. The function you wish to integrate must be stored in the Function Memory as f_6 before you execute this program. You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N).

```
"A="?→A
"B="?→B
"N="?→N
(B-A)÷N→D
0→S
A→X
For 1→K To N
S+f6D→S
X+D→X
Next
A→X
f6 →E
B→X
f6 →F
S+(F-E)D÷2→S
S
```

Simpson's Rule

The following program uses Simpson's Rule to approximate the value of a definite integral. The function you wish to integrate must be stored in the variable f_6 before you execute this program. You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N). You must choose an even number for N. You can also use the calculator's built-in numerical integration program, which uses Simpson's Rule as its algorithm. See the Owner's Manual for more information on the built-in function.

```
"A="?→A
"B="?→B
"N="?→N
(B-A)÷N → D
0 → S
A → X
f6 → S
B → X
S+f6 → S
For 1→K To N÷2
A+(2K-1)D → X
S+4f6 → S
Next
For 1→K To (N÷2)-1
A+2KD → X
S+2f6 → S
Next
SD÷3
```

Integration by Endpoints or Midpoint

The following programs approximate the value of a definite integral. The function you wish to integrate must be stored in the Function Memory as f_6 before you execute this program. You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N).

Left Endpoints:

```
"A="?→A
"B="?→B
"N="?→N
(B-A)÷N→D
0→S
A→X
For 1→K To N
S+f6D→S
X+D→X
Next
S
```

Right Endpoints:

```
"A="?→A
"B="?→B
"N="?→N
(B-A)÷N→D
0→S
A→X
X+D→X
For 1→K To N
S+f6D→S
X+D→X
Next
S
```

Midpoints:

```
"A="?→A
"B="?→B
"N="?→N
(B-A)÷N→D
0→S
A→X
X+.5D→X
For 1→K To N
S+f6D→S
```

```
X+D→X
Next
S
```

Euler's Method

The following program performs Euler's method to approximate the solution to the differential equation

$$\frac{dy}{dx} = f(x).$$

The function $f(x)$ must be stored as $\mathtt{f}_6$ before you execute this program. You also must enter an initial point (both x and y coordinates), the increment between successive x values, and the value of x at which you wish to find an estimate for y.

```
"INITIAL X VALUE"?→X
"INITIAL Y VALUE"?→Y
"INCREMENT"?→H
"FINAL X VALUE"?→Z
Lbl 1
Y+H× f6 → Y
X+H → X
X<Z ⇒ Goto 1
"FINAL X VALUE"
X ▲
"FINAL Y VALUE"
Y
```

Normal Probability

The following program computes the probability that a normal random variable with given mean and standard deviation lies between two given values. The probability that the random variable lies below a particular value may be computed by inputting -1E99 for the lower limit; the probability that the random variable lies above a particular value may be computed by inputting 1E99 for the upper limit.

```
"MEAN"?→M
"STD DEV"?→S
"LOWER LIMIT"?→A
(A-M)÷S→A
If A<-10
```

```
Then -10→A
IfEnd
If A>10
Then 10→A
IfEnd
"UPPER LIMIT"?→B
(B-M)÷S→B
If B<-10
Then -10→B
IfEnd
If B>10
Then 10→B
IfEnd
"PR(A<N(M,S)<B)="
Fix 4
∫(e(Xˆ2÷-2),A,B)÷√(2×π)
```

Programs for the HP-38G

Introduction

The following section contains programs for the HP-38G calculator. These programs allow the calculator to do various financial calculations, linear programming, numerical integration, Euler's method for the numerical solution of differential equations, and to find probabilities associated with the normal distribution.

Entering the Programs Manually: All editing of programs is done within the Program catalog; to enter the Program catalog, press PROGRAM (the blue version of the 0 key). To enter a new program, press the menu key below NEW on the calculator screen; to edit an existing program, use the arrow keys to move down the list of programs until the one you wish to edit is shaded, then press the menu key below EDIT on the calculator screen. If you are creating a new program, you will be prompted for a program name. After entering a name and pressing OK twice, you will be ready to begin entering the program.

There are several places to go to find the characters you will need to enter the programs. Some arithmetic commands are available from the keyboard, as are the alphabet keys. Upper-case letters are entered by pressing the A...Z key (which is just above the blue key on the keypad) then the desired letter; lower-case letters are entered as upper-case, except you press the blue key before beginning the process. To lock the calculator in upper-case alphabet entry mode, you press the menu key below A...Z on the calculator screen; to lock the calculator in lower-case entry mode, press the blue key then the menu key below A...Z. You can easily tell when the calculator is in alphabet entry mode by the presence of an α at the very top of the screen. You can tell when it is locked in alphabet mode by the presence of a square in the A...Z menu key. The ▶ symbol you will need for these programs is the menu key below STO▶ on the calculator screen; keys to enter a space or to backspace are located at the menu keys below SPACE and BKSP on the calculator screen.

In addition to the keyboard symbols, you will need special characters like quotation marks and the summation sign. You will find these symbols by pressing CHARS (the blue version of the (key). Use the arrow keys to shade the symbol you want, then press ENTER; that symbol will be placed into the program at the point where the cursor was last. For commands like INPUT, ROUND, and MSGBOX, you may either enter them letter by letter or by pressing MATH. To find mathematical

commands like ROUND, highlight the Real menu, then move over to the right hand side of the box and move down the list until you highlight ROUND. To find data input and output commands, press the menu key below CMDS on the calculator screen. Move down the left hand side of the box until you highlight Prompt, then move to the right hand side of the box and move down the list until you higlight the appropriate command. Pressing ENTER will return the command to the program at the cursor. The User's Guide contains the menu locations of all available commands.

When reading the programs in the following section, you will notice a seemingly arbitrary amount of horizontal space in the MSGBOX lines of the programs. This space is not arbitrary, but is designed to make the output of the programs to be as readable as possible. You may estimate the number of spaces necessary to fill these gaps, and enter them into the programs. To exit the Program catalog, you may press HOME to return to the home screen.

Running the Programs: You may run the programs from either the home screen or from the Program catalog. To run a program from the catalog, highlight its name using the arrow keys then press the menu key below RUN on the calculator screen. To run a program from the home screen, type RUN followed by the program name. What you see will depend on from where you have run the program. If you run the program from the home screen, the final output of the program will be returned to the screen and will stay there even after the program has terminated and you have pressed OK to return to the home screen. If you run the program from the Program catalog, the output will disappear after you press OK, and you will be returned to the Program catalog screen.

Trapezoidal Rule

The following program uses the Trapezoidal Rule to approximate the value of a definite integral. The function you wish to integrate must be stored in the variable F1(X) before you execute this program. To do this, press the LIB key, note that Function is highlighted, and press ENTER. You may now enter the function F1(X). You are also required to input the lower limit of integration (A), the upper limit of integration (B), and the number of subintervals (N).

```
INPUT A;
"TRAPEZOIDAL RULE";
"A";
"ENTER LOWER LIMIT";
Ø:
```

```
INPUT B;
"TRAPEZOIDAL RULE";
"B";
"ENTER UPPER LIMIT";
Ø:
INPUT N;
"TRAPEZOIDAL RULE";
"N";
"ENTER NUMBER OF INTERVALS";
Ø:
(B-A)/N▶D:
F1(A)+F1(B)▶S:
S+2∗∑(I=1,N-1,F1(A+I∗D))▶S:
MSGBOX "Trapezoidal Rule Sum:          "S∗D/2:
FREEZE:
```

Simpson's Rule

The following program uses Simpson's Rule to approximate the value of a definite integral. The function you wish to integrate must be stored in the variable `F1(X)` before you execute this program. To do this, press the LIB key, note that `Function` is highlighted, and press ENTER. You may now enter the function `F1(X)`. You are also required to input the lower limit of integration (`A`), the upper limit of integration (`B`), and the number of subintervals (`N`). You must choose an even number for `N`.

```
INPUT A;
"SIMPSON'S RULE";
"A";
"ENTER LOWER LIMIT";
Ø:
INPUT B;
"SIMPSON'S RULE";
"B";
"ENTER UPPER LIMIT";
Ø:
INPUT N;
"SIMPSON'S RULE";
"N";
"ENTER (EVEN) NUMBER OF INTERVALS";
Ø:
```

```
(B-A)/N▶D:
F1(A)+F1(B)▶S:
S+4∗∑(I=1,N/2,F1(A+(2∗I-1)∗D)▶S:
S+2∗∑(I=1,(N/2)-1,F1(A+2∗I∗D)▶S:
MSGBOX "Simpson's Rule Sum:                    "S∗D/3:
FREEZE:
```

Integration by Endpoints or Midpoint

The following programs approximate the value of a definite integral. The function you wish to integrate must be stored in the variable `F1(X)` before you execute this program. To do this, press the LIB key, note that `Function` is highlighted, and press ENTER. You may now enter the function `F1(X)`. You are also required to input the lower limit of integration (`A`), the upper limit of integration (`B`), and the number of subintervals (`N`).

Left Endpoints:

```
INPUT A;
"LEFT ENDPOINT SUM";
"A";
"ENTER LOWER LIMIT";
Ø:
INPUT B;
"LEFT ENDPOINT SUM";
"B";
"ENTER UPPER LIMIT";
Ø:
INPUT N;
"LEFT ENDPOINT SUM";
"N";
"ENTER NUMBER OF INTERVALS";
Ø:
(B-A)/N▶D:
∑(I=Ø,N-1,F1(A+I∗D)▶S:
MSGBOX "Left Endpoint Sum:                    "S∗D:
FREEZE:
```

Right Endpoints:

```
INPUT A;
"RIGHT ENDPOINT SUM";
"A";
"ENTER LOWER LIMIT";
Ø:
INPUT B;
"RIGHT ENDPOINT SUM";
"B";
"ENTER UPPER LIMIT";
Ø:
INPUT N;
"RIGHT ENDPOINT SUM";
"N";
"ENTER NUMBER OF INTERVALS";
Ø:
(B-A)/N▶D:
∑(I=1,N,F1(A+I∗D)▶S:
MSGBOX "Right Endpoint Sum:   "S∗D:
FREEZE:
```

Midpoints:

```
INPUT A;
"MIDPOINT SUM";
"A";
"ENTER LOWER LIMIT";
Ø:
INPUT B;
"MIDPOINT SUM";
"B";
"ENTER UPPER LIMIT";
Ø:
INPUT N;
"MIDPOINT SUM";
"N";
"ENTER NUMBER OF INTERVALS";
Ø:
(B-A)/N▶D:
∑(I=1,N,F1(A+(I-.5)∗D)▶S:
MSGBOX "Midpoint Sum:   "S∗D:
FREEZE:
```

Euler's Method

The following program performs Euler's method to approximate the solution to the differential equation

$$\frac{dy}{dx} = f(x).$$

The function $f(x)$ must be stored as `F1(X)` before you execute this program. To do this, press the LIB key, note that `Function` is highlighted, and press ENTER. You may now enter the function `F1(X)`. You also must enter an initial point (both x and y coordinates), the increment between successive x values, and the value of x at which you wish to find an estimate for y.

```
INPUT X;
"EULER'S METHOD";
"XØ";
"ENTER INITIAL X VALUE";
Ø:
INPUT Y;
"EULER'S METHOD";
"YØ";
"ENTER INITIAL Y VALUE";
Ø:
INPUT H;
"EULER'S METHOD";
"INCR";
"ENTER INCREMENT";
Ø:
INPUT Z;
"EULER'S METHOD";
"FINAL X";
"ENTER FINAL X VALUE";
Ø:
WHILE X<Z REPEAT
Y+H*F1(X)▶Y:
X+H▶X:
END:
DISP 2; "FINAL X VALUE:":
DISP 3; " "X:
DISP 4; "FINAL Y VALUE:":
```

```
DISP 5; " "Y:
FREEZE:
```

Normal Probability

The following program computes the probability that a normal random variable with given mean and standard deviation lies between two given values. The probability that the random variable lies below a particular value may be computed by inputting -1E99 for the lower limit; the probability that the random variable lies above a particular value may be computed by inputting 1E99 for the upper limit.

```
INPUT M;
"NORMAL PROBABILITY";
"MEAN";
"ENTER MEAN";
Ø:
INPUT S;
"NORMAL PROBABILITY";
"STD DEV";
"ENTER STANDARD DEVIATION";
Ø:
INPUT A;
"NORMAL PROBABILITY";
"A";
"ENTER LOWER LIMIT";
Ø:
INPUT B;
"NORMAL PROBABILITY";
"B";
"ENTER UPPER LIMIT";
Ø:
UTPN(M,Sˆ2,A)-UTPN(M,Sˆ2,B)▶V:
MSGBOX "Pr(A<N(M,S)<B)=" V:
FREEZE:
```